Alle Dateien eines Ordners in einer Excel-Tabelle ausgeben
Versionen: Excel 97, 2000, 2002/XP und 2003

 FRAGE:  Kann man die Dateien eines Festplattenordners als Verzeichnisliste in eine Excel-Tabelle einfügen? Nach Möglichkeit sollen neben dem Dateinamen auch die Dateigröße sowie Erstelldatum und letztes Änderungsdatum ausgegeben werden.
E. Berger

 ANTWORT:  Für die Aufgabe benötigen Sie ein VBA-Makro. Wie ein solches Makro aussehen könnte, demonstriert das folgende Beispiel:

1. Wechseln Sie aus Excel mit Alt+F11 in den Visual Basic-Editor.

2. Rufen Sie im Menü Einfügen den Befehl Modul auf.

3. In das daraufhin geöffnete Codefenster geben Sie die Prozedur zur Ausgabe der Verzeichnisliste ein:

Sub Dateiliste()
Dim strOrdner As String
Dim fso As Object
Dim fsOrdner As Object
Dim fsDateien As Object, fsDatei As Object
Dim wb As Workbook, sh As Worksheet
Dim i As Long

  strOrdner = "C:\Windows"

  Set fso = CreateObject("Scripting.FileSystemObject")
  If Not fso.FolderExists(strOrdner) Then
    MsgBox "Ordner nicht vorhanden."
    Exit Sub
  Else
    Set fsOrdner = fso.GetFolder(strOrdner)
    Set fsDateien = fsOrdner.Files
  End If

  Set wb = Application.Workbooks.Add
  Set sh = wb.Worksheets(1)

  With sh
    .Cells(3, 1) = "Dateiname"
    .Cells(3, 2) = "Dateigröße"
    .Cells(3, 3) = "Dateityp"
    .Cells(3, 4) = "Erstelldatum"
    .Cells(3, 5) = "Letzter Zugriff"
    .Cells(3, 6) = "Letzte Änderung"
    .Rows(3).Font.Bold = True

    On Error Resume Next
    i = 4
    If fsDateien.Count > 0 Then
      For Each fsDatei In fsDateien
        .Cells(i, 1) = fsDatei.Name
        .Cells(i, 2) = fsDatei.Size
        .Cells(i, 3) = fsDatei.Type
        .Cells(i, 4) = fsDatei.DateCreated
        .Cells(i, 5) = fsDatei.DateLastAccessed
        .Cells(i, 6) = fsDatei.DateLastModified
        i = i + 1
      Next fsDatei
    Else
      .Cells(i, 1) = "Keine Dateien"
    End If
    On Error GoTo 0

    .Columns("A:F").AutoFit
    .Cells(1, 1) = "Inhalt von " & strOrdner
    .Cells(1, 1).Font.Bold = True
    .Cells(1, 1).Font.Size = 12
  End With

  Set fsDatei = Nothing
  Set fsDateien = Nothing
  Set fsOrdner = Nothing
  Set fso = Nothing
  Set sh = Nothing
  Set wb = Nothing
End Sub

· In der vorliegenden Fassung erhalten Sie beim Ausführen des Makros ein Inhaltsverzeichnis des Ordners "C:\Windows". Wenn es diesen Ordner auf Ihrem System gar nicht gibt oder wenn Sie lieber den Inhalt eines anderen Ordners ausgeben wollen, passen Sie die Pfadangabe in der Makrozeile

strOrdner = "C:\Windows"

an Ihre Systemumgebung an.

· Verlassen Sie den Visual Basic-Editor über das Menü Datei und den Befehl Schließen und zurück zu Microsoft Excel.

Wenn Sie das Makro gleich ausprobieren wollen, drücken Sie im Excel-Fenster die Tastenkombination Alt+F8. Damit öffnen Sie ein Dialogfenster mit einer Liste verfügbarer Makros. Doppelklicken Sie auf den Makronamen "Dateiliste".

Das Makro legt daraufhin eine neue Arbeitsmappe an und fügt die Dateinamen inklusive Dateigröße, Dateityp und verschiedener Datumsangaben in das erste Tabellenblatt dieser Mappe ein.

Der Komfort des Makros lässt sich sicherlich noch verbessern. Deshalb hier ein paar Hinweise zur generellen Funktionsweise:

Ganz am Anfang übergeben Sie den Ordner, dessen Dateien in eine Excel-Tabelle eingetragen werden sollen, an die Variable "strOrdner"; hier in Form des festen Textwertes "C:\Windows".

Auf die Ordnerinhalte greifen Sie anschließend mit Hilfe des "FileSystemObject" zurück. Den Objektbezug auf den Ordner erhalten Sie mit der "GetFolder"-Methode und die Dateien des Ordners finden Sie in der "Files"-Auflistung des Ordnerobjekts.

Als nächstes bereiten Sie die neue Arbeitsmappe vor, in der die Dateiliste ausgegeben werden soll. In die dritte Zeile des ersten Tabellenblatts schreiben Sie die Spaltenüberschriften "Dateiname", "Dateigröße" usw.

Anschließend durchlaufen Sie in einer For-Schleife alle Dateien des Ordners. Dabei werden Name, Größe ("Size"), Dateityp ("Type") usw. in die Excel-Tabelle eingetragen.

Am Ende nehmen Sie noch kleinere Schönheitskorrekturen an der Excel-Tabelle vor, indem Sie die Breite der Spalten A bis F auf die optimale Breite einstellen und in der ersten Zeile eine Überschrift mit Hinweis auf den durchsuchten Ordner einfügen. Und um den durch die Objektvariablen belegten Arbeitsspeicher wieder frei zu geben, setzen Sie in den letzten Makrozeilen alle Objekte auf "Nothing".

Aufbauend auf diesen Hinweisen können Sie den Makrocode beliebig erweitern, - beispielsweise um einen Ordner auf Unterverzeichnisse zu prüfen (Unterverzeichnisse erhalten Sie über die "SubFolders"-Auflistung des Ordnerobjekts).
